Modeling Balloon Diffusion Dynamics in MLAB

Gary D. Knott, Ph.D.
Civilized Software, Inc.
12109 Heritage Park Circle
Silver Spring MD 20906
Tel.: (301)-962-3711
email: csi@civilized.com
URL: www.civilized.com

abstract: An inflated balloon that is shrinking due to loss of inside-air due to diffusion across the balloon wall is modeled with the MLAB mathematical modeling system.

At time 0, we are given a spherical balloon, initially of interior radius $r(0)$ inches (and interior volume $v(0)$) and containing $m(0)$ pounds of air (at a higher density than that of the outside air.) The thickness of the rubber wall of the balloon at time 0 is $h(0)$ inches. The air in the balloon *diffuses* uniformly across the rubber balloon wall at a rate directly proportional to the pressure difference between the inside air and the outside air, and inversely proportional to the thickness of the rubber balloon wall. [Is it true that the diffusion rate is simply directly proportional to the pressure difference?] At time t, the interior radius is $r(t)$, the interior volume is $v(t)$, the inside-air mass is $m(t)$, and the balloon wall thickness is $h(t)$.

The ideal gas law states that pressure is proportional to temperature times density, and, in our case, temperature may be assumed to be constant, so then pressure and density are proportional, where density equals mass/volume. Thus $p(t) = \gamma m(t)/v(t)$, where $p(t)$ is the inside pressure at time t.

As the inside air diffuses out of the balloon, the balloon shrinks in size. This loss of air reduces the amount, and hence the mass, of the inside air remaining, and, together with the shrinking, also affects the inside pressure, $p(t)$. The shrinking of the balloon is governed by the reduction of stress in the balloon wall due to the lower pressure causing it.

We wish to know the time-course values of the balloon radius function, $r(t)$, as the balloon shrinks to its equilibrium state, and we also want to know the functions $p(t)$, $m(t)$, and $h(t)$.

Let h_0 be the thickness of the unstressed balloon wall and let r_0 be the inside radius of the unstressed balloon. Then, assuming the density of the rubber material remains unchanged under stress, the thickness, $h(t)$, at time t satisfies $\frac{4}{3}\pi[(r(t)+h(t))^3 - r(t)^3] = \frac{4}{3}\pi[(r_0+h_0)^3 - r_0^3]$. (It may be more realistic to assume the density of the rubber wall material decreases as it is stressed, and it would be straightforward to accomodate this extension.)

Thus we have:
inside mass of air at time t: $m(t)$
inside radius of balloon at time t: $r(t)$
balloon wall thickness at time t: $h(t)$
inside balloon volume at time t: $v(t) = \frac{4}{3}\pi r(t)^3$
inside balloon surface area at time t: $s(t) = 4\pi r(t)^2$
outside balloon surface area at time t: $4\pi(r(t)+h(t))^2$
inside pressure at time t: $p(t) = \gamma m(t)/v(t)$
constant outside pressure: u
net pressure difference: $d(t) = p(t) - u$
unstressed balloon inside radius: r_0
unstressed balloon inside surface area: $s_0 = 4\pi r_0^2$
unstressed balloon wall thickness: h_0
diffusion coefficient per unit area: $\alpha d(t)/h(t)$

The net diffusion rate is: $\frac{dm(t)}{dt} = -\alpha(d(t)/h(t))s(t)m(t)$, with $m(0) = m_0$.

Note we could replace $s(t)$ in our diffusion differential equation with the more correct "midway" surface area that lies between the inside surface area and the outside surface area; this would accomodate the slightly greater escape probability compared to the case of diffusion across a planar membrane.

A small nearly-rectangular patch of unstressed balloon wall can be modeled by a vertical spring and a horizontal spring, so that, by Hooke's law, the force needed to stretch an $L \times L$ unstressed patch horizontally to size $x \times L$ is approximately $\varepsilon(x-L)/L$, and independently, the force needed to stretch such a patch vertically to height x is also approximately $\varepsilon(x-L)/L$, where ε is the non-directional modulus of elasticity of our rubber membrane material.

The total force $F_L(x)$ needed to stretch an unstressed $L \times L$ patch to an $x \times x$ patch is thus $F_L(x) = 2\varepsilon(x - L)/L$. (In real situations, the modulus of elasticity may appear to be a function of x, especially as x becomes large and our material approaches its elastic limit.) Now for a complete spherical balloon having the unstressed inside surface area $s_0 = 4\pi r_0^2$, a uniformly-applied force of $2\varepsilon(s^{1/2} - s_0^{1/2})/s_0^{1/2}$ will stretch the balloon to have the inside surface area s.

The force F on the entire rubber membrane is just the inside area times the inside pressure minus the outside area times the outside pressure; this is approximately the inside area times the pressure difference $F = 4\pi r(t)^2 d(t)$. This can be obtained by an alternate approach due to Jay Lindau. Consider *half* our inflated balloon, *i.e.* a hemisphere, placed in coodinatized space so that the hemisphere rests on the *xy*-plane and spherically bulges in the z-direction. The disk in the *xy*-plane defined by the hemisphere will be called the base of the hemisphere. If we consider outward vectors of force normal to the hemisphere surface representing the forces due to the pressure difference that is keeping our rubber balloon membrane stretched, then the *sum* of all these vectors cancel all components of these vectors except the components normal to the z-direction. Thus the area of the base of the hemisphere times the pressure difference equals the z-component force, and hence the total force, on the hemisphere rubber membrane, *i.e.* $F_z = d(t)2\pi r(t)^2$. The force on the entire rubber membrane is thus again $F = 4\pi r(t)^2 d(t)$.

Since "stress" is (approximately) proportional to "strain" in a slowly-deflating balloon, we may write:

$$2\varepsilon(s(t)^{1/2} - s_0^{1/2})/s_0^{1/2} = 4\pi r(t)^2 d(t) = s(t)d(t).$$

Thus, $2\varepsilon s(t)^2 = s_0^{1/2}(2\varepsilon + s(t)d(t))$. And $s(t) = 4\pi r(t)^2$ and $d(t) = \gamma m(t)/(\frac{4}{3}\pi r(t)^3) - u$, so:

$$r(t)^4 - s_0^{1/2}[\frac{3\gamma}{2\varepsilon(4\pi)^2}\frac{m(t)}{r(t)} - \frac{ur(t)^2}{2\varepsilon 4\pi} + \frac{1}{(4\pi)^2}] = 0.$$

Note the dynamic behavior of our balloon depends on the constants r_0, h_0, ε, α, u, γ, and either p_0 or m_0 (since they are related.)

If we have time-course information about r and/or m and/or p, we may estimate parameters such as ε, α, γ, or u by curve-fitting. Indeed, estimating u is equivalent to calibrating a barometer. (Note we have a particularly nice barometer when $\alpha \approx 0$.)

Below is an MLAB do-file for computing and graphing the time-course dynamics of a fictional balloon obeying the above equations.

```
/* file: balloon.do - define and compute the time-course dynamics
   of a slowly-deflating balloon. */
reset
echodo=3

p0=28; u=14; r0=1; h0=.01; g=.5; a=.0001; e=.1

fct v(r)=4*pi*r^3/3
fct s(r)=4*pi*r^2

ml=v(r0)*p0/g; s0=s(r0)

/* mh()= inside-air mass for pressure p0 */
fct mh() = root(m0,ml,100*ml,g*m0/v(r(m0))-p0)

/* m't(t) = differential-equation for inside-air mass which changes
   due to diffusion. */
fct m't(t) = -a*f(r(m),m)*m
init m(0) = m0

/* r(m) = balloon radius for inside-air mass m */
fct r(m) = root(b,r0/1.2,m*r0/u,b^4-c1*m/b+c2*b^2-c3)
c1=sqrt(s0)*3*g/(2*e*(4*pi)^2)
c2=sqrt(s0)*u/(2*e*4*pi)
c3=sqrt(s0)/(4*pi)^2

/* p(m,r) = pressure for inside-air mass m in a sphere of radius r */
fct p(r,m) = g*m/v(r)

fct f(r,m) = ((p(r,m)-u)*s(r))/h(r)

/* h(r) = balloon wall thickness for a balloon of radius r */
fct h(r) = root(z,0,r0,(r+z)^3-r^3-hv)
hv=(r0+h0)^3-r0^3

m0=mh() /* compute initial value of m */
```

```
tv=0:6!160   /* table of time values */

q=points(m,tv) /* q= values of m on tv paired with these tv-values */

rv=r on (q col 2)  /* rv= radius values for given inside-air masses */

draw tv&'rv  lt dashed                    /* r */
frame 0 to .5, 0 to .5
top title "r vs. t"
w1=w

draw q                                    /* m */
frame 0 to .5, .5 to 1
top title "m vs. t"
w2=w

draw tv&'(p on (rv&'(q col 2))) color red /* p */
frame .5 to 1, 0 to .5
top title "p vs. t"
w3=w

draw tv&'(h on rv) color green            /* h */
frame .5 to 1, .5 to 1
top title "h vs. t"
w4=w

view
```



problem: Model the dynamics of an inflated spherical balloon that is punctured with a pin. Why does a punctured balloon make a 'pop' sound? Why does the balloon sometimes rupture into torn pieces?